数学实验第五次实验报告

计算机系　计 43　2014011330　黄家晖

2017 年 4 月 21 日

1　实验目的

- 掌握使用 LINGO 解线性规划的方法；
- 通过求解实际问题，学会建立实际问题的线性规划模型。

2　应用题

2.1　CH8-6 证券投资策略选择

2.1.1　模型建立

设公司购进了 A 证券 x_A 万元，B 证券 x_B 万元，C 证券 x_C 万元，D 证券 x_D 万元，E 证券 x_E 万元，考虑纳税情况，则总收益为：

$$z = p_A x_A + \frac{1}{2} p_B x_B + \frac{1}{2} p_C x_C + \frac{1}{2} p_D x_D + p_E x_E$$

其中，$p_A, \cdots, p_E$ 代表各个证券的税前收益。系数 $\frac{1}{2}$ 代表 50% 的纳税。

约束条件 (1) 在上述模型下可以表达为：

$$x_B + x_C + x_D \geq 400$$

约束条件 (2) 可以表达为：

$$\frac{l_A x_A + l_B x_B + l_C x_C + l_D x_D + l_E x_E}{x_A + x_B + x_C + x_D + x_E} \leq 1.4$$

化简为：

$$(1.4 - l_A)x_A + (1.4 - l_B)x_B + (1.4 - l_C)x_C + (1.4 - l_D)x_D + (1.4 - l_E)x_E \geq 0$$

其中，$l_A, \cdots, l_E$ 代表各个证券的税前收益。

约束条件 (3) 可以表达为：

$$\frac{y_A x_A + y_B x_B + y_C x_C + y_D x_D + y_E x_E}{x_A + x_B + x_C + x_D + x_E} \leq 5$$

化简为：

$$(5-y_A)x_A+(5-y_B)x_B+(5-y_C)x_C+(5-y_D)x_D+(5-y_E)x_E \geq 0$$

其中，$y_A, \cdots, y_E$ 代表各个证券的税前收益。

对于第 (1) 问，需要增加条件：

$$x_A+x_B+x_C+x_D+x_E \leq 1000$$

对于第 (2) 问，假设借用的金额为 k 万元，需要增加条件：

$$k \leq 100$$

且需要修正目标函数为：

$$z=p_A x_A+\frac{1}{2}p_B x_B+\frac{1}{2}p_C x_C+\frac{1}{2}p_D x_D+p_E x_E-0.0275k$$

同时增加约束：

$$x_A+x_B+x_C+x_D+x_E \leq k+1000$$

2.1.2 算法设计

根据上述模型分析可知，所有约束均为线性不等式，目标函数也为线性，符合线性规划的一般形式，因此使用 LINGO 的 Linear Programming 功能直接进行求解即可。

对于第 (3) 问的目标函数参数敏感性分析，可以通过 LINGO 的 `Prices & Ranges`命令进行分析。

2.1.3 LINGO 模型

第一问代码如下

```
MODEL:
Title Stock Investment;
sets:
  sn/1..5/: p, l, y, x, cutoff;
endsets
data:
  p = 0.043 0.054 0.05 0.044 0.045;
  l = 2 2 1 1 5;
  y = 9 15 4 3 2;
  cutoff = 1 0.5 0.5 0.5 1;
enddata
[OBJECTIVE] max = @sum(sn: cutoff * p * x);
[CONS1] x(2) + x(3) + x(4) >= 400;
[CONS2] @sum(sn: (1.4 - l) * x) >= 0;
[CONS3] @sum(sn: (5 - y) * x) >= 0;
[T1CONS] @sum(sn: x) <= 1000;
END
```

第二问代码如下

```
MODEL:
Title Stock Investment;
sets:
  sn/1..5/: p, l, y, x, cutoff;
```

```
5 endsets
6 data:
7   p = 0.043 0.054 0.05 0.044 0.045;
8   l = 2 2 1 1 5;
9   y = 9 15 4 3 2;
10  cutoff = 1 0.5 0.5 0.5 1;
11 enddata
12 [OBJECTIVE] max = @sum(sn: cutoff * p * x) — 0.0275 * k;
13 [CONS1] x(2) + x(3) + x(4) >= 400;
14 [CONS2] @sum(sn: (1.4 — l) * x) >= 0;
15 [CONS3] @sum(sn: (5 — y) * x) >= 0;
16 [T2CONS1] @sum(sn: x) <= 1000 + k;
17 [T2CONS2] k <= 100;
18 END
```

2.1.4 计算结果与分析

下面分析的条件行号均对应上一节代码中定义的行号。

第 (1) 问 LINGO 输出的结果使用了 LP 模型，这与预期符合。求解器迭代次数为 3，最终结果如表 1所示。

变量	最优值	减少费用
x_A	218.18	0（基变量）
x_B	0.0	0.03
x_C	736.36	0（基变量）
x_D	0.0	0.0006
x_E	45.45	0（基变量）

表 1: 第一问求解结果

另外，根据 LINGO 的松弛变量输出，还可知条件 CONS2、CONS3、T1CONS的松弛变量为 0，说明这几个约束起了作用。

最优目标函数值为 29.836。

第 (2) 问 LINGO 输出的结果使用了 LP 模型，求解器迭代次数为 3，最终结果如表 2所示。

变量	最优值	减少费用
x_A	240.0	0（基变量）
x_B	0.0	0.03
x_C	810.0	0（基变量）
x_D	0.0	0.0006
x_E	50.0	0（基变量）
k	100	0（基变量）

表 2: 第二问求解结果

另外，根据 LINGO 的松弛变量输出，还可知条件 CONS2、CONS3、T2CONS1、T2CONS2的松弛变量为 0，说明这几个约束起了作用。

最优目标函数值为 30.07。

第 (3) 问 针对第 (1) 问的 Range 分析，可以得到的分析如下：

```
                  Objective Coefficient Ranges:

                   Current        Allowable        Allowable
 Variable      Coefficient         Increase         Decrease
    X( 1)        0.4300000E-01    0.3500000E-02    0.1300000E-01
    X( 2)        0.2700000E-01    0.3018182E-01     INFINITY
    X( 3)        0.2500000E-01    0.1733333E-01    0.5600000E-03
    X( 4)        0.2200000E-01    0.6363636E-03     INFINITY
    X( 5)        0.4500000E-01    0.5200000E-01    0.1400000E-01
```

当证券 A 的税前收益增加 0.2% 的时候，在变量 `X(1)`的增加允许范围 0.35% 之内，所以投资可以不改变；当证券 C 税前收益减少 0.2% 的时候，不在变量 `X(3)`的减少允许范围 0.056% 之内，所以投资应该改变。

问题解答 若该经理有 1000 万元资金，应该投资 A 证券 218.18 万元、投资 C 证券 736.36 万元、投资 E 证券 45.45 万元，其他证券不投资，可获取的最大收益为 29.84 万元。

若能够借贷 100 万元资金，应该借贷 100 万元、投资 A 证券 240 万元、投资 C 证券 810 万元、投资 E 证券 50 万元，其他证券不投资，可获取的最大收益为 30.07 元。

若证券 A 税前收益增加为 4.5%，投资不应该改变；当证券 C 税前收益减少为 4.8% 时，投资应该改变。

2.2 CH8-9 果仁混合问题

2.2.1 模型建立

为方便叙述，给普通、豪华和蓝带糖编号为成品 1，2 和 3；给杏仁、核桃仁、腰果仁和胡桃仁编号为原料 1，2，3 和 4。设每周商店为成品 i 购买的原材料 j 为 a_{ij} 千克（$i = 1, 2, 3; j = 1, 2, 3, 4$）。

设成品售价向量为：

$$s = [0.89 \quad 1.10 \quad 1.80]^T$$

设原料售价向量为：

$$g = [0.45 \quad 0.55 \quad 0.70 \quad 0.50]$$

则商店每周的利润为：

$$z = \Sigma_i s_i \Sigma_j a_{ij} - \Sigma_j g_j \Sigma_i a_{ij}$$

题目中给出的约束条件有：

$$\Sigma_i a_{i1} \leq 2000$$

$$\Sigma_i a_{i2} \leq 4000$$

$$\Sigma_i a_{i3} \leq 5000$$

$$\Sigma_i a_{i4} \leq 3000$$

关于材料数量的限制条件为：

$$a_{13} \leq 0.2\Sigma_j a_{1j} \quad a_{14} \geq 0.4\Sigma_j a_{1j} \quad a_{12} \leq 0.25\Sigma_j a_{1j}$$

$$a_{23} \leq 0.35\Sigma_j a_{2j} \quad a_{21} \geq 0.4\Sigma_j a_{2j}$$

$$a_{31} \geq 0.3\Sigma_j a_{3j} \quad 0.3\Sigma_j a_{3j} \leq a_{33} \leq 0.5\Sigma_j a_{3j}$$

2.2.2 算法设计

根据上述模型分析可知，所有约束均为线性不等式，目标函数也为线性，符合线性规划的一般形式，因此使用 LINGO 的 Linear Programming 功能直接进行求解即可。

2.2.3 LINGO 模型

```
1 MODEL:
2 Title Sugar Manufactoring;
3 sets:
4   ns/1..3/: s;
5   ng/1..4/: g, totalcons;
6   link(ns, ng): A;
7 endsets
8 data:
9   s = 0.89 1.10 1.80;
10  g = 0.45 0.55 0.70 0.50;
11  totalcons = 2000 4000 5000 3000;
12 enddata
13 [OBJECTIVE] max = @sum(ns(i): s(i) * @sum(ng(j): a(i, j))) - @sum(ng(j): g(j) * @sum(ns(i): a(i, j)));
14 ! The number of gradient <= threshold;
15 @for(ng(j):
16   [GRADIENT_NUM] @sum(ns(i): a(i, j)) <= totalcons(j);
17 );
18 ! Category-specific gradient constraints;
19 [CATE_SPEC1] a(1, 3) <= 0.2 * @sum(ng(j): a(1, j));
20 [CATE_SPEC2] a(1, 4) >= 0.4 * @sum(ng(j): a(1, j));
21 [CATE_SPEC3] a(1, 2) <= 0.25 * @sum(ng(j): a(1, j));
22 [CATE_SPEC4] a(2, 3) <= 0.35 * @sum(ng(j): a(2, j));
23 [CATE_SPEC5] a(2, 1) >= 0.4 * @sum(ng(j): a(2, j));
24 [CATE_SPEC6] a(3, 1) >= 0.3 * @sum(ng(j): a(3, j));
25 [CATE_SPEC7] a(3, 3) >= 0.3 * @sum(ng(j): a(3, j));
26 [CATE_SPEC8] a(3, 3) <= 0.5 * @sum(ng(j): a(3, j));
27 END
```

2.2.4 计算结果与分析

下面分析的条件行号均对应上一节代码中定义的行号。

最优值	$j=1$	$j=2$	$j=3$	$j=4$
$i=1$	0	1363.63	1090.91	3000.0
$i=2$	0	0	0	0
$i=3$	2000.0	2636.36	2030.30	0

表 3: 果糖店各个 a_{ij} 的最优值

减少费用	$j=1$	$j=2$	$j=3$	$j=4$
$i=1$	3.32	0（基变量）	0（基变量）	0（基变量）
$i=2$	0（基变量）	1.78	1.78	2.12
$i=3$	0（基变量）	0（基变量）	0（基变量）	0.35

表 4: 果糖店各个 a_{ij} 的减少费用

计算结果 各个变量的最优值如表 3所示，他们的减少费用如表 4所示。

题目中所给的各个条件松弛值如下：

```
            Row    Slack or Surplus      Dual Price
      OBJECTIVE         10069.70          1.000000
GRADIENT_NUM( 1)        0.000000          3.916667
GRADIENT_NUM( 2)        0.000000         0.1500000
GRADIENT_NUM( 3)        1878.788          0.000000
GRADIENT_NUM( 4)        0.000000         0.5454545
     CATE_SPEC1         0.000000         0.3454545
     CATE_SPEC2         818.1818          0.000000
     CATE_SPEC3         0.000000         0.3454545
     CATE_SPEC4         0.000000          0.000000
     CATE_SPEC5         0.000000         -5.444444
     CATE_SPEC6         0.000000         -3.666667
     CATE_SPEC7         30.30303          0.000000
     CATE_SPEC8         1303.030          0.000000
```

可以看出有 4 个条件对决定最优值没有影响。最终模型计算出的最优值为 10069.70。

问题解答 商店每周应购进杏仁 2000 千克，核桃仁 4000 千克，腰果仁 3121.21 千克，胡桃仁 3000 千克；生产普通品牌糖果 5454.54kg，生产蓝带品牌糖果 6666.67kg，豪华糖果不生产。周利润可达到 1.007 万美元。

具体来说，2000 千克杏仁全部用来制作蓝带；4000 千克核桃仁分 1363.63 千克制作普通、2636.36 制作蓝带；3121.21 千克腰果仁分 1090.91 千克制作普通、2030.30 千克制作蓝带；3000 千克胡桃仁全部制作普通糖果。

2.3 CH8-10 污水处理问题

2.3.1 模型建立

设工厂总个数为 n，对应居民区总数也为 n，设每个居民区上游水流量为 F_i，下游水流量为 T_i，上游污水浓度为 U_i，下游污水浓度为 D_i，从第 i 个居民区到第 $i+1$ 个居民区中间段江水的自净系数为 c_i。第 i 个工厂的排水量为 P_i，处理前污水浓度为 b_i，处理过后的污水浓度为 d_i，处理系数为 e_i。则可以写出如下等式（$i=1,2,\cdots n$）：

$$F_i = T_{i-1}$$

$$T_i = F_i + P_i$$

$$U_i = (1 - c_{i-1})D_{i-1}$$

$$D_i = \frac{F_i U_i + P_i d_i}{F_i + P_i}$$

并特别给出：

$$F_1 = L_0 \quad U_1 = a_0$$

其中 L_0 和 a_0 是江水初始的流量和污水浓度。

为了满足实际污水处理情况，还需给出：

$$b_i \geq d_i, \quad i = 1, 2, \cdots n$$

江面上所有地段水污染指数最高值为：

$$\max_i D_i$$

所有居民点上游水污染指数的最高值为：

$$\max_i U_i$$

如果给出最高值的限制 ϵ，则上述条件可以扩展为 n 个限制（这一步转换是等价的）：

$$D_i \leq \epsilon, \quad i = 1, 2, \cdots n$$

或

$$U_i \leq \epsilon, \quad i = 1, 2, \cdots n$$

实际上，建立上述模型以及得出最高值的相关结论是基于对模型的一定假设和近似的，具体对于模型的假设如下：

- 假设污水产生的根源仅仅在于工厂；
- 假设工厂 i 和居民区 i 在江水对岸，及工厂 i 的上游也是居民区 i 的上游，工厂 i 的下游也是居民区 i 的下游；
- 假设工厂在上游和下游的交界处瞬间将污水处理完成，且排出的污水和江水瞬间混合；
- 假设在工厂处理污水的时候，不改变污水的流量。

最后，处理污水总共需要花费的费用为：

$$\theta = \Sigma_{i=1}^{n} e_i P_i (b_i - d_i)$$

在处理具体问题的时候，由于上述变量大多为已知，带入具体数值即可，所要优化的变量只有 d_i，可以近似成线性规划问题求解。

2.3.2 算法设计

根据上述模型分析可知，所有约束均为线性不等式，目标函数也为线性，符合线性规划的一般形式，因此使用 LINGO 的 Linear Programming 功能直接进行求解即可。本题的难点在于模型的建立。

2.3.3 LINGO 模型

最小化江面所有地段水污染的模型为：

```
MODEL:
Title River Pollution;
sets:
  nd/1..3/: b, d, dd;
endsets
data:
  b = 100 60 50;
enddata
[OBJECTIVE] min = 5 * @sum(nd: b - d);
dd(1) = (1000 * 0.8 + 5 * d(1)) / 1005.0;
dd(2) = (1005 * 0.9 * dd(1) + 5 * d(2)) / 1010.0;
dd(3) = (1010 * 0.6 * dd(2) + 5 * d(3)) / 1015.0;
dd(1) <= 1;
dd(2) <= 1;
dd(3) <= 1;
b(1) >= d(1);
b(2) >= d(2);
b(3) >= d(3);
END
```

最小化居民点上游的污水浓度模型为：

```
MODEL:
Title River Pollution;
sets:
  nd/1..3/: b, d, dd;
endsets
data:
  b = 100 60 50;
enddata
[OBJECTIVE] min = 5 * @sum(nd: b - d);
dd(1) = (1000 * 0.8 + 5 * d(1)) / 1005.0;
dd(2) = (1005 * 0.9 * dd(1) + 5 * d(2)) / 1010.0;
dd(3) = (1010 * 0.6 * dd(2) + 5 * d(3)) / 1015.0;
0.9 * dd(1) <= 1;
0.6 * dd(2) <= 1;
b(1) >= d(1);
b(2) >= d(2);
b(3) >= d(3);
END
```

2.3.4 计算结果与分析

计算结果 当限制江面所有地段水质的时候，模型求解变量的最优值为：

$$d_1 = 41.0 \quad d_2 = 21.1 \quad d_3 = 50.0$$

对应的减少费用均为 0，代表他们均为基变量。实际上，对于本小问，对应的约束矩阵是满秩的。

最少费用为 489.5 万元。

当限制居民点上游水质的时候，模型求解变量的最优值为：

$$d_1 = 63.3 \quad d_2 = 60.0 \quad d_3 = 50.0$$

对应的减少费用均为 0，代表他们均为基变量。

最少费用为 183.33 万元。

结果分析 对比两种模型，仅处理居民上游点水质的时候，费用明显减少。实际上，在给定的假设下，限制所有地段水质等价于既处理居民点上游水质、又处理居民点下游的水质。第二种方案充分受益于河水的自净机制，可想而知如果自净系数为 1（即没有自净功能）或接近 1，这两种方案所需的花费差距将大大减小。

问题解答 当限制江面所有地段水质的时候，最少需要花费 489.5 万元；当限制居民点上游水质的时候，最少需要花费 183.33 万元。

3 收获与建议

通过这次的实验，我学会了 LINGO 软件的基本使用和编程技巧，并对线性规划有了更深的理解。希望在之后的课堂上老师能够当堂进行相关的技巧演示并给出题目的分步解答。