数学实验第七次实验报告

计算机系　计 43　2014011330　黄家晖

2017 年 5 月 9 日

1　实验目的

- 掌握使用 LINGO 解整数规划模型的方法；
- 通过求解实际问题，学会建立实际问题的整数规划模型。

2　应用题

2.1　CH10-6 二次指派问题

2.1.1　模型建立

设有分配矩阵 A，大小为 $n \times n$，每个元素 $a_{ij} \in \{0, 1\}$ 代表员工到城市的分派情况。当 $a_{ij} = 0$ 的时候代表员工 i 没有被分配到城市 j；当 $a_{ij} = 1$ 的时候代表员工 i 被分配到城市 j。

设员工之间通话时长矩阵 B，大小为 $n \times n$，每个元素 b_{ij} 表示员工 i 和员工 j 的每月通话时长；设有城市之间费率矩阵 C，大小为 $n \times n$，每个元素 c_{ij} 表示城市 i 和城市 j 的通话费率。

对于一个分配矩阵 A，显然需要保证每个员工仅仅被分配到一个城市，同时也要保证每个城市都仅仅分得一个员工：

$$\Sigma_j a_{ij} = 1, \quad \forall i$$

$$\Sigma_i a_{ij} = 1, \quad \forall j$$

设通话费率为 $f(A)$，则可以由上述符号表示为：

$$f(A) = \Sigma_i \Sigma_j a_{ij} \Sigma_{k,i<k} \Sigma_{m,m \neq j} a_{km} b_{ik} c_{mj}$$

上述式子各个字母的含义是：i 和 k 分别为员工 1 和员工 2 的编号，为了保证费率计算不重复（即两个员工相互之间的通话费用只计算一份），需要要求 $i < k$；而 j 和 m 表示员工 1 和员工 2 所在的城市，使用的求和符号的原因是：由于 a_{ij} 在不存在分配关系的情况下是 0，相当于没有相应的那一项。

因此我们的优化目标为：

$$\min f(A)$$

2.1.2 算法设计

分析上述优化问题的各个变量，可以发现需要求解的 A 矩阵为 0-1 矩阵，因此可以采用 LINGO 软件自带的 **0-1 整数规划**来进行问题求解。

2.1.3 LINGO 模型

```
model:
sets:
n/1..10/:;
nsq(n, n): assign, ft;
endsets
data:
ft = 0 5 3 7 9 3 9 2 9 0
7 0 7 8 3 2 3 3 5 7
4 8 0 9 3 5 3 3 9 3
6 2 10 0 8 4 1 8 0 4
8 6 4 6 0 8 8 7 5 9
8 5 4 6 6 0 4 8 0 3
8 6 7 9 4 3 0 7 9 5
6 8 2 3 8 8 6 0 5 5
6 3 6 2 8 3 7 8 0 5
5 6 7 6 6 2 8 8 9 0;
enddata
min = @sum(n(i): ! Employer;
  @sum(n(j): ! City;
    assign(i, j) * @sum(n(k) | i #LT# k: ! Other employer;
      @sum(n(m) | m #NE# j: ! Other city;
        assign(k, m) * ft(i, k) * @if(j #GT# m, ft(j, m), ft(m, j));
      )
    )
  )
);
@for(nsq: @bin(assign));
@for(n(i): @sum(n(j): assign(i, j)) = 1);
@for(n(j): @sum(n(i): assign(i, j)) = 1);
end
```

2.1.4 计算结果与分析

LINGO 最终的输出结果如下：

```
 Global optimal solution found.
 Objective value:                              1142.000
 Objective bound:                              1142.000
 Infeasibilities:                              0.000000
 Extended solver steps:                              83
 Total solver iterations:                         33085

 Model Class:                                     PINLP

 Total variables:                   100
 Nonlinear variables:               100
 Integer variables:                 100

 Total constraints:                  21
 Nonlinear constraints:               1

 Total nonzeros:                    300
```

```
 Nonlinear nonzeros:                 100

       Variable           Value
...
  ASSIGN( 1, 9)        1.000000
...
  ASSIGN( 2, 1)        1.000000
...
  ASSIGN( 3, 8)        1.000000
...
  ASSIGN( 4, 3)        1.000000
...
  ASSIGN( 5, 6)        1.000000
...
  ASSIGN( 6, 7)        1.000000
...
  ASSIGN( 7, 2)        1.000000
...
  ASSIGN( 8, 5)        1.000000
...
  ASSIGN( 9, 4)        1.000000
...
 ASSIGN( 10, 10)       1.000000
```

根据 LINGO 的输出结果可以看出，每个月花费的最小值为 1142 元。进行该问题的求解耗时月 3 分钟左右，可见有了 100 个待求变量的整数规划问题在进行求解的时候比较耗时。程序采用的是 B-B 分支定界法，如果在搜索的时候能够加入一些启发信息或是在条件中加入一些额外的约束则可以加快搜索的过程。

问题解答 员工 1 应分到城市 9，员工 2 应分到城市 1，员工 3 应分到城市 8，员工 4 应分到城市 3，员工 5 应分到城市 6，员工 6 应分到城市 7，员工 7 应分到城市 2，员工 8 应分到城市 5，员工 9 应分到城市 4，员工 10 应分到城市 10。每月最小费用为 1142.0 元。

2.2 CH10-9 原油采购与加工

2.2.1 模型建立

假设原油 A 使用存货中的 at 进行加工甲、乙两种汽油，又购进新原油 A 总共 tt，而最终在生产时，甲汽油含有原油 A 总共 et，乙汽油含有原油 A 总共 ft；原油 B 中使用 ct 加工甲汽油，使用 dt 加工乙汽油。根据题目中的约束，可以列出下列条件：

$$a \leq 500, \quad t \geq 0$$

$$c + d \leq 1000$$

$$e + f = a + t$$

$$e \geq 0.5(e + c), \quad f \geq 0.6(f + d)$$

需要注意的是，原油 A 的使用上，并没有添加“必须先使用存货”的限制，这是因为最终的最优结果一定会倾向于使用存货，因为不会花费多于的财产，对结果没有影响。

考虑到新购进的原油价格随着购买吨数的不同有着不同的价格，其曲线为分段函数的形状，因此考虑将其分成三段 t_1, t_2, t_3，分别代表购进吨数 500 吨以内的部分，超过 500 吨但不超过 1000 吨的部分，以及超过 1000 吨的部分。考虑到实际购买情况，列出该分段函数相关的下列约束关系：

$$t = t_1 + t_2 + t_3$$

$$t_1 \leq 500, \quad t_2 \leq 500, \quad t_3 \leq 500$$

$$t_2(t_1 - 500) = 0$$

$$t_3(t_2 - 500) = 0$$

最终的盈利最大即为优化目标，如下所示：

$$\max z = 4800(e + c) + 5600(f + d) - 10000t_1 - 8000t_2 - 6000t_3$$

2.2.2 算法设计

题目要求使用连续规划和整数规划两种模型来求解这个问题，因此可以使用 LINGO 的 `@gin`关键字为相关的变量添加整数优化，从而求解本问题。

2.2.3 LINGO 模型

```
model:
a <= 500;
e >= 0.5 * (e + c);
f >= 0.6 * (f + d);
c + d <= 1000;
e + f = a + t;
t = t1 + t2 + t3;
t1 <= 500;
t2 <= 500;
t3 <= 500;
t2 * (t1 - 500) = 0;
t3 * (t2 - 500) = 0;
max = 4800 * (e + c) + 5600 * (f + d) - 10000 * t1 - 8000 * t2 - 6000 * t3;
end
```

为了求解整数规划问题，所需要的模型修改如下：

```
model:
a <= 500;
e >= 0.5 * (e + c);
f >= 0.6 * (f + d);
c + d <= 1000;
e + f = a + t;
t = t1 + t2 + t3;
t1 <= 500;
t2 <= 500;
t3 <= 500;
t2 * (t1 - 500) = 0;
```

```
t3 * (t2 - 500) = 0;
@gin(a); @gin(e); @gin(c); @gin(f); @gin(d);
@gin(t1); @gin(t2); @gin(t3);
max = 4800 * (e + c) + 5600 * (f + d) - 10000 * t1 - 8000 * t2 - 6000 * t3;
end
```

2.2.4 计算结果与分析

使用连续规划求解的结果如下：

$$a = 500.0, \quad e = 0.0, \quad c = 0.0$$

$$f = 1500.0, \quad d = 1000.0, \quad t = 1000.0$$

最优值为 5,000,000.0。

使用整数规划求解的结果如下：

$$a = 500, \quad e = 0, \quad c = 0$$

$$f = 1500, \quad d = 1000, \quad t = 1000$$

最优值为 5,000,000。

对比连续规划和整数规划能够看出，二者求出的结果完全相同，对应的最优值也相同，但是就求解过程来看，整数规划使用的是 MINLP 模型，迭代轮数为 938；而连续规划使用的是 NLP 模型，迭代轮数为 206，远远少于整数规划，但最终结果也基本为整数。可以看出，整数规划在求解方面的确效率不如普通的连续规划。

另外，实际的最终结果表明不需要生产甲汽油。这说明了单单对比甲乙两种汽油的产出投入比，显然乙汽油能够花费少的原料产生更多的利润。尽管生产单位的甲汽油所需要的原油 A 较少，但是其卖出的售价也相应的低很多，利润赶不上乙汽油。

问题解答 该公司应该购买原油 A 总共 1000t，并且使用所有的原油 A 和原油 B 全部用来生产乙汽油，能够使得利润最大，为 5 百万元。

2.3 CH10-11 钢管下料

2.3.1 模型建立

假设使用了这四种切割模式，他们切割的钢管数目分别为 x_1, x_2, x_3, x_4，且保证 $x_1 \geq x_2 \geq x_3 \geq x_4$。并设第 i 种切割模式下每根原料钢管生产长 290mm, 315mm, 350mm, 455mm 钢管各 $r_{1i}, r_{2i}, r_{3i}, r_{4i}$ 根。

为了生产满足用户需求数目的钢管，需要的约束条件为：

$$x_1 r_{11} + x_2 r_{12} + x_3 r_{13} + x_4 r_{14} \geq 15$$

$$x_1 r_{21} + x_2 r_{22} + x_3 r_{23} + x_4 r_{24} \geq 28$$

$$x_1 r_{31} + x_2 r_{32} + x_3 r_{33} + x_4 r_{34} \geq 21$$

$$x_1r_{41} + x_2r_{42} + x_3r_{43} + x_4r_{44} \geq 30$$

$$\Sigma_i r_{ij} \leq 5, \forall j$$

题目中给出了余料限制为*每种切割模式下的余料浪费不能超过 100mm*，这句话有两种理解：

- 理解一：对于第 i 种切割方式切割的所有钢管余料之和不超过 100mm；
- 理解二：对于第 i 种切割方式每个钢管的余料不超过 100mm。

对于理解一，首先保证每根钢管剩余的长度在 0 到 290mm 之间：

$$1560 \leq 290r_{11} + 315r_{21} + 350r_{31} + 455r_{41} \leq 1850$$

$$1560 \leq 290r_{12} + 315r_{22} + 350r_{32} + 455r_{42} \leq 1850$$

$$1560 \leq 290r_{13} + 315r_{23} + 350r_{33} + 455r_{43} \leq 1850$$

$$1560 \leq 290r_{14} + 315r_{24} + 350r_{34} + 455r_{44} \leq 1850$$

其次，对于所有钢管的余料之和有相关限制：

$$x_1(1850 - (290r_{11} + 315r_{21} + 350r_{31} + 455r_{41})) \leq 100$$

$$x_2(1850 - (290r_{12} + 315r_{22} + 350r_{32} + 455r_{42})) \leq 100$$

$$x_3(1850 - (290r_{13} + 315r_{23} + 350r_{33} + 455r_{43}) \leq 100$$

$$x_4(1850 - (290r_{14} + 315r_{24} + 350r_{34} + 455r_{44})) \leq 100$$

对于理解二，需要对理解一的前四个条件进行更改：

$$1750 \leq 290r_{11} + 315r_{21} + 350r_{31} + 455r_{41} \leq 1850$$

$$1750 \leq 290r_{12} + 315r_{22} + 350r_{32} + 455r_{42} \leq 1850$$

$$1750 \leq 290r_{13} + 315r_{23} + 350r_{33} + 455r_{43} \leq 1850$$

$$1750 \leq 290r_{14} + 315r_{24} + 350r_{34} + 455r_{44} \leq 1850$$

为了使总费用最少，即是使得购买钢管所花费的钱数和加工所需要的钱数总和最小，假设一根钢管的价值为 1，即优化目标为：

$$\min z = 1.1x_1 + 1.2x_2 + 1.3x_3 + 1.4x_4$$

2.3.2 算法设计

上述模型涉及到的变量 x 和 r 均是整数，且涉及到两个优化变量的乘法，可以使用 LINGO 的 PINLP 模型进行求解，注意必须使用破解版，否则试用版不提供该模型。

2.3.3 LINGO 模型

针对理解一：

```
model:
sets:
  nn/1..4/: req, w, x, len;
  link(nn, nn): r;
endsets
data:
  w = 1.1 1.2 1.3 1.4;
  req = 15 28 21 30;
  len = 290 315 350 455;
enddata
min = @sum(nn: w * x);
@for(nn(i):
  @sum(nn(j): x(j) * r(i, j)) >= req(i);
  @sum(nn(j): r(j, i)) <= 5;
  @sum(nn(j): len(j) * r(j, i)) >= 1560;
  @sum(nn(j): len(j) * r(j, i)) <= 1850;
  x(i) * (1850 - @sum(nn(j): len(j) * r(j, i))) <= 100;
  @gin(x(i));
  @for(nn(j): @gin(r(i, j)));
);
end
```

针对理解二：

```
model:
sets:
  nn/1..4/: req, w, x, len;
  link(nn, nn): r;
endsets
data:
  w = 1.1 1.2 1.3 1.4;
  req = 15 28 21 30;
  len = 290 315 350 455;
enddata
min = @sum(nn: w * x);
@for(nn(i):
  @sum(nn(j): x(j) * r(i, j)) >= req(i);
  @sum(nn(j): r(j, i)) <= 5;
  @sum(nn(j): len(j) * r(j, i)) >= 1750;
  @sum(nn(j): len(j) * r(j, i)) <= 1850;
  @gin(x(i));
  @for(nn(j): @gin(r(i, j)));
);
end
```

2.3.4 计算结果与分析

计算结果 对于理解一，经过求解器的 121 万轮迭代，发现不存在最优解（`No feasible solution found.`）。

对于理解二，计算得到的最优切割方式如表 1所示。

	290mm	315mm	350mm	455mm
第一种切割方式	1	2	0	2
第二种切割方式	0	0	5	0
第三种切割方式	2	0	1	2
第四种切割方式	0	3	1	1

表 1: 最优解切割方式

且：

$$x_1 = 14, \quad x_2 = 4, \quad x_3 = 1, \quad x_4 = 0$$

可以的得到最小费用为 21.5。

由求解结果可以看出，第四种切割方式实际上并没有被使用，所以得出的第四种切割方式实际上也没有意义，在给出最终结果的时候应该舍去。

问题解答 对于理解一，最优化问题无解，没有结论。

对于理解二，为了使总费用最小，应该采用表 2所示的三种切割方式。使用第一种切割方式购进并切割 14 根钢管、使用第二种方式购进并切割 4 根钢管、使用第三种方式购进并切割 1 根钢管，即可以在满足用户需求的前提下花费最少的费用，所得到的最小费用为 21.5 倍个钢管价值。

	290mm	315mm	350mm	455mm
第一种切割方式	1	2	0	2
第二种切割方式	0	0	5	0
第三种切割方式	2	0	1	2

表 2: 实际采用的三种切割方式

3 收获与建议

通过这次的实验，我学会了 LINGO 软件的基本使用和编程技巧，并对整数规划有了更深的理解。希望在之后的课堂上老师能够当堂进行相关的技巧演示并给出题目的分步解答。